

Dwarf Boards


DN009 : read 5 analog inputs, show on LCD, in Jal on DB016 on DB016

(c) Van Ooijen Technische Informatica
version 1.0

PICmicro, In-Circuit Serial Programming and ICSP are registered trademarks of Microchip Technology Inc.

Introduction

This note shows how read 4 analog inputs and show the values on an LCD.



Code

```
-- DN009 : read bus_a_pin_0..3 as analog, show on LCD on bus_c
-- hardware: DB016 + DB011 on bus_a + 2x16 LCD on BUS_C

include DB016_18F252_PLL_40
include jdelay
```

The first include is specific for the board, microcontroller and frequency used, it establishes among other things the I/O declarations that are used. The next include establishes the delay declarations.

```
const LCD_lines = 2
include lcd_on_c
```

The next two lines establish the LCD interface.

```
bus_a_direction = all_input
bus_c_direction = all_output
```

The BUS_A is made input (required to use these pins as analog inputs). BUS_C (used for the LCD) is made output.

```
function hex( byte in x ) return byte is
  if x < 10 then
    return "0" + x
  else
    return ( "A" + x ) - 10
  end if
end function
```

The function hex returns the ASCII character corresponding to the value of the argument x. X must be in the range 0 .. 15, otherwise an 'invalid' value will be returned.

```
procedure show_channel( byte in channel ) is
  var byte result_1, result_0
  adc_read_10( channel, result_1, result_0 )
  LCD_Write( hex( result_1 & 0b_11 ) )
  LCD_Write( hex( result_0 >> 4 ) )
  LCD_Write( hex( result_0 & 0b_1111 ) )
end procedure
```

The procedure show_channel reads the channel indicated by the argument, and writes the value to the LCD as three hexadecimal characters.

```
procedure show( byte in n, byte in channel ) is
  LCD_Write( hex( n ) )
  LCD_Write( ":" )
  show_channel( channel )
end procedure
```

The procedure show writes the number n, a colon, and calls show_channel to write the value of the A/D channel.

```
LCD_Init
LCD_Clear
adc_configure( adc_5_ana_0_ref )
forever loop

  LCD_Line_1
  show( 0, adc_bus_a_pin_0 )
  LCD_Write( " " )
  show( 1, adc_bus_a_pin_1 )

  LCD_Line_2
  show( 2, adc_bus_a_pin_2 )
  LCD_Write( "," )
  show_channel( adc_bus_a_pin_3 )
  LCD_Write( " " )
  show( 5, adc_bus_a_pin_5 )

  delay_100ms
end loop
```

The main code first initializes and clears the LCD, and initializes the A/D. The main loop selects line 1, and calls show twice to display A/D pins 0 and 1. Next line 2 is selected and show is called to write A/D pin 2. Pins 2 and 3 are both connected to the LM35, so after pin 2 a comma is written and show_channel is called directly (instead of via show) to display pin 3. Finally pin 5 (pin 4 is does not support analog functions) is shown, delay_100ms is called and the main loop starts again.

The values shown for pin 0 and 1 can be controlled between 000 and 3FF by turning the two potentiometers on the DB011 board.

Typical values shown for the LM35 pins are 0BC (pin 2, LM35 zero) and 0F0 (pin 3, LM35 output). This corresponds to an LM35 output of 34 hex = 52 decimal. The analog reference is the +5 Volt supply, so 52 corresponds with $(52 / 1023) * 5 \text{ Volt} = 0.254 \text{ Volt}$, or 25.4 degrees Celcius. Indeed, it is a bit warm here.

Typical values for pin 5 (LCD) are 360 when directly under a lamp, 2F0 in normal office light, 090 when covered with a dark sheet of paper.

Change notes

the latest version of this document can be downloaded from <http://www.voti.nl/dwarf>

| version | date | notes |
|---------|------------|---------------|
| 1.0 | 2003-11-04 | first version |